


GIOCO DEL TRIS



CmdEND: serve a chiudere il programma; utilizza la procedura

```
Private Sub cmdEND_Click()  
    'chiudo il programma  
    End  
End Sub
```

CmdSTART: serve a inizializzare il gioco quando ricomincio dopo la fine di una partita; utilizza la procedura

```
Private Sub cmdSTART_Click()  
    Dim i As Integer  
    If nm = 9 Then  
        'pulisco il campo di gioco  
        For i = 0 To 8  
            campo(i).Caption = ""  
        Next i  
        'azzerare il numero delle mosse  
        nm = 0  
        'simboli  
        simboli = "XO"  
        'vittoria  
        vinto = False  
    End If  
End Sub
```

FormLoad: stabilisce le condizioni iniziali del gioco all'apertura della form

```
Private Sub Form_Load()  
    nm = 9  
    cmdSTART_Click  
End Sub
```

Espressioni globali: stabilisce le condizioni globali del gioco

```
Option Explicit  
'variabili globali  
Dim simboli As String  
Dim vinto As Boolean 'indica la vittoria  
Dim nm As Integer 'numero della mossa
```

Funzione Vittoria: stabilisce il vincitore del gioco, quando c'è

```
Function vittoria(i As Integer) As Boolean
    Dim r, c As Integer
    Dim x, y As Integer
    Dim ng As Integer
    Dim s As String
    Dim flag As Boolean
    Dim conta As Integer

    flag = False
    ng = nm Mod 2
    s = Mid(simboli, ng + 1, 1)
    r = Int(i / 3)
    c = Int(i Mod 3)
    If r = c Then
        'diagonale principale
        conta = 0
        For x = 0 To 2
            If campo(x * 3 + x).Caption = s Then conta = conta + 1
        Next x
        If conta = 3 Then flag = True
    End If
    If r = 2 - c Then
        'diagonale secondaria
        conta = 0
        For x = 0 To 2
            If campo(x * 3 + (2 - x)).Caption = s Then conta = conta + 1
        Next x
        If conta = 3 Then flag = True
    End If
    'riga
    conta = 0
    For y = 0 To 2
        If campo(r * 3 + y).Caption = s Then conta = conta + 1
    Next y
    If conta = 3 Then flag = True
    'colonna
    conta = 0
    For x = 0 To 2
        If campo(x * 3 + c).Caption = s Then conta = conta + 1
    Next x
    If conta = 3 Then flag = True
    vittoria = flag
End Function
```

CampoClick: gestisce le fasi del gioco

```
Private Sub campo_Click(Index As Integer)
    Dim gioca As Integer
    'se la cella è vuota
    If campo(Index).Caption = "" And nm < 9 Then
        'inserisco il simbolo del giocatore
        gioca = nm Mod 2
        campo(Index).Caption = Mid(simboli, gioca + 1, 1)
        'guardo se il giocatore corrente ha vinto
        vinto = vittoria(Index)
        If vinto = True Then
            MsgBox "Il giocatore " & _
                Mid(simboli, gioca + 1, 1) & _
                " ha vinto la partita!"
            nm = 8
        End If
        'incremento il conteggio delle mosse
        nm = nm + 1
    End If
    If nm = 9 And vinto = False Then
        MsgBox "Partita pari!"
    End If
End Sub
```